Tema 8
Cálculo diferencial de funciones reales de variable real (2)

Objetivos

1. Cálculo de los extremos locales de una función con wxMaxima.
2. Cálculo de los extremos absolutos de una función en un conjunto.
3. Conocer e interpretar los teoremas del valor medio.
4. Aplicar las reglas de l'Hôpital.
5. Calcular el desarrollo de Taylor de una función derivable.
6. Calcular el desarrollo en serie de potencias de una función.
7. Aplicación práctica de las propiedades de las funciones derivables.

Contenidos

Referencias

AEM11 ALANINOS PRATS, J; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P. (2011)
Cálculo con wxMaxima.

APJ11 ALANINOS PRATS, J; APARICIO DEL PRADO, C; EXTREMERA LIZANA, J; MUÑOZ RIVAS, P.; VILLENA MUÑOZ, A.R. (2011)
Prácticas de ordenador con wxMaxima.

AP86 APOSTOL, T.M. (1986)
Análisis Matemático

BR09 BRUZÓN GALLEGO, M. DE LOS SANTOS; RAMÍREZ LABRADOR, JOSÉ (2009)
Modelos matemáticos con Maxima

BU07 DE BURGOS, JUAN (2007)
Cálculo Infinitesimal de una variable (segunda edición).

ES08 ESTELA CARBONELL, M. ROSA; SAÀ SEOANE, JOEL (2008)
Cálculo con soporte interactivo en moodle.

ES02 ESTEP, DONALD (2002)
Practical Analysis in one variable

GV07a GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 1: números y ecuaciones

GV07b GONZÁLEZ VEGA, LAUREANO (2007)
Laboratorio de Matemáticas. Vol. 2: límites y derivadas

JB01 JARAUTA BRAGULAT, EUSEBI (2001)
Anàlisi Matemàtica d'una variable. Fonaments i aplicacions.

RR05 REDONDO NEBLE, M. VICTORIA; RODRÍGUEZ GALVÁN, J. RAFAEL (2005)
Introducción a Maxima

RR08b RODRÍGUEZ RIOTORTO, MARIO (2008)
Curso intensivo i-MATH de software libre orientado a Ciencias e Ingeniería

RU80 RUDIN, WALTER (1980)
Principios de Análisis Matemático.

SP95 SPIVAK, MICHAEL (1995)
Calculus (Càlcul Infinitesimal).

VR09 VALLEJO RODRÍGUEZ, JOSÉ ANTONIO (2009)
Cálculo diferencial con Maxima

08-1.- Extremos locales de una función. Extremos absolutos

Los contenidos de este apartado se desarrollan en el archivo **Tema_08-1.wxm**.

Definición (extremos locales o relativos).

Consideremos una función real de variable real $f:[a,b]\subset\mathbb{R}\to\mathbb{R}$; entonces:

- se dice que f tiene un *máximo local* (o máximo relativo) en el punto $x_0\in[a,b]$, si existe $r>0$ tal que
$$f(x_0)\geq f(x),\ \ \forall x\in B(x_0;r)\cap[a,b]$$
- se dice que f tiene un *mínimo local* (o máximo relativo) en el punto $x_0\in[a,b]$, si existe $r>0$ tal que
$$f(x_0)\leq f(x),\ \ \forall x\in B(x_0;r)\cap[a,b]$$

Los máximos y mínimos locales se llaman *extremos locales o relativos*. La interpretación gráfica de esta definición se puede ver a continuación.

Máximo local de una función derivable:

```
(%i1) f1(x):=2 - (x - 1)^2$
      wxplot2d([f1(x)], [x,0,3], [y, 0, 3], [ylabel, ""] , [style, [lines, 2]] , [legend, "f1(x)"],
      [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 200] )$
plot2d: some values were clipped.
```

(%t2)



Mínimo local de una función derivable:

```
(%i3) f2(x):=x^2-2*x+2$
      wxplot2d([f2(x)], [x,0,3], [y, 0,3], [ylabel, ""] , [style, [lines, 2]] , [legend, "f1(x)"],
      [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 200] )$
plot2d: some values were clipped.
```

(%t4)



Mínimo local de una función no derivable:

```
(%i5) f3(x):=abs(x^2-2*x-8)$
      wxplot2d([f3(x)], [x, 2, 6] , [y, -2, 6], [ylabel, ""] , [style, [lines, 2]] , [legend, "f1(x)"],
      [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 200] )$
plot2d: some values were clipped.
```

(%t6)



Proposición. Si una función f tiene un extremo local en un punto x_0 y es derivable en este punto, entonces se cumple $Df(x_0) = 0$.

Este resultado significa que la recta tangente a la gráfica de la función es horizontal, tal como se muestra a continuación.

Máximo local de una función derivable:

```
(%i7) f1(x):=2 - (x - 1)^2$       y1(x):=2$
      wxplot2d([f1(x), y1(x)], [x,0,3], [y, 0,3], [ylabel, ""] , [style, [lines, 2]] , [legend, "f1(x)", "y1(x)"],
      [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 200] )$
plot2d: some values were clipped.
```

(%t9)



Mínimo local de una función derivable:

```
(%i10) f2(x):=x^2-2*x+2$        y2(x):=1$
       wxplot2d([f2(x), y2(x)], [x,0,3], [y, 0,3], [ylabel, ""] , [style, [lines, 2]] , [legend, "f2(x)", "y2(x)"],
       [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 200] )$
plot2d: some values were clipped.
```

(%t12)



Los puntos en los que se cumple la propiedad $Df(x)=0$ se denominan *puntos críticos* de la función. La proposición anterior dice que todo extremo local es un punto crítico. El recíproco no es cierto, como ilustra por ejemplo la función $f(x)=x^3$, que cumple $Df(0)=0$ y sin embargo no presenta un extremo local en este punto.

Punto crítico de una función derivable en el que no hay extremo local:

```
(%i13) f4(x):=x^3$          y4(x):=0$
       wxplot2d([f4(x), y4(x)], [x,-2,2], [ylabel, ""] , [style, [lines, 2]] , [legend, "f4(x)", "y4(x)"],
       [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 200] )$
```

(%t15)



Estudio de los extremos locales de una función.

Consideremos una función $f: A \subset \mathbb{R} \to \mathbb{R}$. Para determinar los extremos locales de la función, hay que estudiar:

(1) El dominio o campo de existencia de la función, el campo de continuidad y el campo de derivabilidad.
(2) Los puntos críticos de la función.
(3) Los puntos en los que la función no es derivable.
(4) El signo de la función derivada en un entorno de los puntos críticos y de los puntos en los que la función no es derivable.

Por ejemplo, en el caso de la función definida por $f(x) = 2-(x-1)^2$, su campo de existencia, de continuidad y de derivabilidad es la recta real. La función derivada es $Df(x) = -2(x-1)$. Determinemos los puntos críticos de la función:

```
(%i1) f1(x):=2 - (x - 1)^2;
(%o1) f1(x):=2-(x-1)^2
(%i2) define (Df1(x)  , diff(f1(x), x) );
(%o2) Df1(x):=-2 (x-1)
(%i3) Eq1:  Df1(x) = 0;    solve(Eq1, x);
(%o3) -2 (x-1)=0
(%o4) [x=1]
```

Por lo tanto, el único punto crítico es $x_0 = 1$. A continuación, estudiamos el signo de la función derivada en un entorno de este punto. Comencemos por la derecha del punto:

```
(%i5) assume(x >0, x < 1);
(%o5) [x>0,x<1]
(%i6) is(Df1(x) > 0);
      is(Df1(x) < 0);
(%o6) true
(%o7) false
```

Ahora la izquierda del punto crítico:

```
(%i8) forget(x >0, x < 1)$
      assume(x >1, x < 2);
(%o9) [x>1,x<2]
(%i10) is(Df1(x) > 0);
       is(Df1(x) < 0);
(%o10) false
(%o11) true
```

Así pues, la derivada a la izquierda del punto crítico es positiva y a la derecha es negativa, por lo que el punto crítico es un máximo local de la función. Si la situación fuera la contraria, entonces se trataría de un mínimo local de la función.

En el caso de la función definida por $f(x) = x^3$, su campo de existencia, de continuidad y de derivabilidad es la recta real. La función derivada es $Df(x) = 3x^2$. Determinamos los puntos críticos de la función:

```
(%i1) f4(x):=x^3;
(%o1) f4(x):=x^3
(%i2) define (Df4(x) , diff(f4(x), x) );
(%o2) Df4(x):=3 x^2
(%i3) Eq2:  Df4(x) = 0;    solve(Eq2, x);
(%o3) 3 x^2=0
(%o4) [x=0]
```

Por lo tanto, el único punto crítico es $x_0 = 0$. Estudiamos ahora el signo de la función derivada en un entorno de este punto:

```
(%i5) assume(x >-1, x < 0);
(%o5) [x>-1,x<0]
(%i6) is(Df4(x) > 0);     is(Df4(x) < 0);
(%o6) true
(%o7) false
```

```
(%i8) forget(x >-1, x < 0)$
      assume(x >0, x < 1);
(%o9) [x>0,x<1]
(%i10) is(Df4(x) > 0);     is(Df4(x) < 0);
(%o10) true
(%o11) false
```

Por lo tanto, la derivada a la izquierda del punto crítico es positiva y a la derecha también es positiva y el punto crítico no es un extremo local de la función.

En el caso de la función definida por $f(x)=\left|x^2-2x-8\right|$, la función no es derivable en el punto $x_1=4$. La función derivada es

$$Df(x)=(2x-2)\frac{x^2-2x-8}{\left|x^2-2x-8\right|},\ x\neq -2,\ x\neq 4\,.$$

Analicemos el cambio de signo de la derivada en el entorno del punto $x_1=4$:

```
(%i1) f3(x):=abs(x^2-2*x-8);
(%o1) f3(x):=|x^2-2 x-8|
(%i2) define (Df3(x) , diff(f3(x), x) );
(%o2) Df3(x):=(2 x-2) (x^2-2 x-8)/|x^2-2 x-8|
(%i3) assume(x > 3, x < 4);
(%o3) [x>3,x<4]
(%i4) is(Df3(x) > 0);     is(Df3(x) < 0);
(%o4) false
(%o5) true
(%i6) forget(x > 3, x < 4)$     assume(x > 4, x < 5);
(%o7) [x>4,x<5]
(%i8) is(Df3(x) > 0);     is(Df3(x) < 0);
(%o8) true
(%o9) false
```

Por lo tanto, la derivada a la izquierda del punto crítico es negativa y a la derecha es positiva y el punto crítico es un extremo local (mínimo).

Definición (extremos absolutos).

Consideremos una función real de variable real $f:[a,b]\subset\mathbb{R}\to\mathbb{R}$; entonces:

- se dice que f tiene el *máximo absoluto* en el punto $x_M\in[a,b]$, si se cumple
$$f(x_M)\geq f(x),\ \ \forall x\in[a,b]$$
- se dice que f tiene el *mínimo absoluto* en el punto $x_m\in[a,b]$, si se cumple
$$f(x_m)\leq f(x),\ \ \forall x\in[a,b]$$

La existencia de extremos absolutos de una función en un intervalo cerrado se determina por aplicación del teorema de Weierstrass, que afirma que la continuidad de la función en un intervalo cerrado garantiza la existencia de extremos absolutos de la función en este intervalo. Naturalmente, puede haber extremos absolutos sin estas hipótesis.

Por lo tanto, el estudio de los extremos (locales y absolutos) de una función $f:[a,b]\subset\mathbb{R}\to\mathbb{R}$ en un intervalo se ha de hacer aplicando la metodología siguiente:

(1) El dominio de la función, el campo de continuidad y el campo de derivabilidad.
(2) Los puntos críticos de la función.
(3) Los puntos en los que la función no es derivable.
(4) Determinar el valor de la función en los extremos a, b del intervalo.

Consideremos por ejemplo la función $f:[-6,8]\subset\mathbb{R}\to\mathbb{R}$ definida por

$$f(x)=\sqrt[3]{x^2(6-x)},\ -6\leq x\leq 8$$

Por aplicación de los criterios de continuidad, esta función es continua en el intervalo considerado y, por lo tanto, tiene extremos absolutos en dicho intervalo. Para estudiar los extremos, determinamos la función derivada, los puntos críticos y los puntos en los que la función no es derivable:

(%i1) f5(x):=(x^2*(6-x))^(1/3);

(%o1) $f5(x):=\left(x^2\,(6-x)\right)^{1/3}$

(%i2) diff(f5(x), x);

(%o2) $\dfrac{2\,(6-x)^{1/3}}{3\,x^{1/3}}-\dfrac{x^{2/3}}{3\,(6-x)^{2/3}}$

(%i3) ratsimp(%);

(%o3) $-\dfrac{x-4}{(6-x)^{2/3}\,x^{1/3}}$

(%i4) define (Df5(x) , %);

(%o4) $Df5(x):=-\dfrac{x-4}{(6-x)^{2/3}\,x^{1/3}}$

Por lo tanto, la función derivada es

$$Df(x) = \frac{4-x}{x^{1/3}(6-x)^{2/3}},\ -6 < x < 8,\ x \neq 0,\ x \neq 6$$

Calculamos ahora los puntos críticos de la función:

```
(%i5) Eq3a:  Df5(x) = 0;     solve(Eq3a, x);
```

$$\text{(\%o5)}\ -\frac{x-4}{(6-x)^{2/3}\,x^{1/3}}=0$$

$$\text{(\%o6)}\ [x=4]$$

El único punto crítico de la función es $x_1 = 4$ ya que la derivada se anula en este punto; la función no es derivable en los puntos $x_2 = 0$ y $x_3 = 6$. Ahora hay que estudiar el signo de la función derivada en un entorno de cada uno de estos puntos. Veamos de menor a mayor y comencemos estudiando el punto $x_2 = 0$:

```
(%i7) assume(x>-1, x<0);
      is(Df5(x) > 0);     is(Df5(x) < 0);
(%o7) [x>-1,x<0]
(%o8) false
(%o9) true
(%i10) forget(x>-1, x<0)$    assume(x>0, x<1);
       is(Df5(x) > 0);     is(Df5(x) < 0);
(%o11) [x>0,x<1]
(%o12) true
(%o13) false
```

Por lo tanto, la función tiene un mínimo local en este punto.

Veamos ahora el punto $x_1 = 4$:

```
(%i14) forget(x>0, x<1)$   assume(x>3, x<4);
       is(Df5(x) > 0);     is(Df5(x) < 0);
(%o15) [x>3,x<4]
(%o16) true
(%o17) false
(%i18) forget(x>3, x<4)$    assume(x>4, x<5);
       is(Df5(x) > 0);     is(Df5(x) < 0);
(%o19) [x>4,x<5]
(%o20) false
(%o21) true
```

Así pues, la función tiene un máximo local en este punto.

En el punto $x_3 = 6$ se cumple:

```
(%i22) forget(x>4, x<5)$  assume(x>5, x<6);
       is(Df5(x) > 0);     is(Df5(x) < 0);
(%o23) [x>5,x<6]
  (%o24) false
  (%o25) true
(%i26) forget(x>5, x<6)$  assume(x>6, x<7);
       is(Df5(x) > 0);     is(Df5(x) < 0);
(%o27) [x>6,x<7]
  (%o28) false
  (%o29) true
```

Por lo tanto, la función no tiene extremo local en este punto. Para determinar los extremos absolutos hay que elaborar una lista con los extremos locales y los extremos del intervalo y calcular el menor y el mayor de todos ellos:

```
(%i30) a=a:-6$   b=b:8$    x1=x1:4$    x2=x2:0$
       [a, x1, x2, b];   [f5(a), f5(x1), f5(x2), f5(b)];
       max(f5(a), f5(x1), f5(x2), f5(b));
       min(f5(a), f5(x1), f5(x2), f5(b));
(%o34) [-6,4,0,8]
```

(%o35) $[3\,2^{4/3}, 2\,4^{1/3}, 0, -2^{7/3}]$

(%o36) $3\,2^{4/3}$

(%o37) $-2^{7/3}$

En definitiva el máximo absoluto está en el punto $a = -6$ y el mínimo absoluto en el punto $b = 8$. Veamos la representación gráfica de la función:

```
(%i38) wxplot2d([f5(x)], [x,-6,8], [ylabel, ""] , [style, [lines, 2]] , [legend, "f5(x)"],
       [gnuplot_preamble, "set size ratio 1; set grid"] , [nticks, 200] )$
```

(%t38)



08-2.- Teoremas del valor medio

Los contenidos de este apartado se desarrollan en el archivo **Tema_08-2.wxm**.

En este apartado se comentan y se ilustran dos teoremas importantes relacionados con la derivada de una función.

Teorema de Rolle. Consideremos una función $f:[a,b]\rightarrow\mathbb{R}$, tal que es continua en el intervalo $[a,b]$ y derivable en el intervalo $]a,b[$. Si se cumple la condición de igualdad de valores de la función en los extremos del intervalo, es decir $f(a)=f(b)$, entonces la función tiene al menos un punto crítico en el intervalo $]a,b[$.

Consideremos por ejemplo la función definida por:

$$f(x)=6-(x+1)(x-2)^2,\ \ 0\leq x\leq 3$$

```
(%i1) f(x):=6-(x+1)*(x-2)^2;
(%o1) f(x):=6-(x+1)(x-2)^2
```

Representación gráfica de la función:

```
(%i2) wxplot2d([f(x)], [x,0,3] , [y,0,9], [ylabel, ""] , [legend, "f(x)"] , [style, [lines, 2]] ,
      [gnuplot_preamble, "set grid"] , [nticks, 200] )$
```

(%t2)



La derivada de esta función es:

$$f(x)=-2(x+1)(x-2)^2-(x-2)^2,\ \ 0<x<3$$

Se cumple $f(0) = f(3) = 0$. Por lo tanto la función tiene un punto crítico en el interior del intervalo, que es el punto $x_0 = 2$. En efecto:

(%i3) define (Df(x) , diff(f(x), x));

(%o3) $Df(x) := -2\,(x-2)(x+1)-(x-2)^2$

(%i4) Eq1: Df(x) = 0;   solve(Eq1, x);

(%o4) $-2\,(x-2)(x+1)-(x-2)^2 = 0$

(%o5) $[x=0, x=2]$

Si la función $f : [a,b] \to \mathbb{R}$ no es derivable en el intervalo abierto $]a,b[$, entonces puede no existir ningún punto en el que se cumpla el teorema. En efecto, consideremos por ejemplo la función definida por:

$$f(x) = 1 + \left| x^2 - 2x - 8 \right|, \; 1+\sqrt{2} \le x \le 5$$

Definición y gráfica de la función:

(%i1) f(x):=1+abs(x^2 - 2*x - 8);

(%o1) $f(x) := 1 + \left| x^2 - 2\,x - 8 \right|$

(%i2) wxplot2d([f(x)], [x,1+sqrt(2), 5], [y,0,10], [ylabel, ""] , [legend, "f(x)"] , [style, [lines, 2]] , [gnuplot_preamble, "set grid"] , [nticks, 200])$

(%t2)



Se cumple la condición de igualdad de valores en los extremos del intervalo, pero la función no es derivable en el intervalo abierto, ya que la función derivada es:

$$f(x) = \frac{(2x-2)(x^2-2x-8)}{\left| x^2 - 2x - 8 \right|}, \; 1+\sqrt{2} < x < 5, \; x \ne 4$$

En efecto, veamos las comprobaciones correspondientes:

```
(%i3) f(1+sqrt(2)), numer;     f(5), numer;
(%o3) 8.0
(%o4) 8
(%i5) define (Df(x) , diff(f(x), x) );
```

$$(\%o5)\ Df(x):=\frac{(2x-2)(x^2-2x-8)}{|x^2-2x-8|}$$

```
(%i6) solve(Df(x)=0, x);
(%o6) [x=-2,x=4,x=1]
(%i7) assume(x>1+sqrt(2), x<4)$
      is(Df(x) < 0);
(%o8) true
(%i9) forget(x>1+sqrt(2), x<4)$
      assume(x>4, x<6)$     is(Df(x) > 0);
(%o11) true
```

Teorema del valor medio de Lagrange (o fórmula del incremento finito).
Consideremos una función $f:[a,b]\to\mathbb{R}$, tal que es continua en el intervalo $[a,b]$ y derivable en el intervalo $]a,b[$. Entonces, existe al menos un punto $x_0\in]a,b[$ tal que:

$$f(b)-f(a)=(b-a)Df(x_0)$$

Interpretación geométrica: la recta tangente a la gráfica de la función en el punto $(x_0, f(x_0))$ es paralela a la cuerda que une los puntos $(a, f(a))$ y $(b, f(b))$. Veamos un ejemplo. Consideremos la función definida por

$$f(x)=(x-1)(x-2)(x-3),\ \ 0\le x\le 5$$

La recta que une los puntos (0,f(0)) i (5, f(5)) tiene por ecuación:

$$y-f(0)=\frac{f(5)-f(0)}{5-0}(x-0)\quad\Leftrightarrow\quad y=6x-6$$

Veamos el detalle de los cálculos y la representación gráfica conjunta de la función y de esta recta:

```
(%i1) f(x):=(x-1)*(x-2)*(x-3);
(%o1) f(x):=(x-1)(x-2)(x-3)
```

(%i2) y - f(0) = (f(5)-f(0)) / (5-0) *(x-0);

(%o2) $y+6=6\,x$

(%i3) define (y(x), 6*x - 6);

(%o3) $y(x):=6\,x-6$

(%i4) wxplot2d([f(x),y(x)], [x,0,5], [y,-10, 30], [ylabel, ""] , [legend, "f(x)" , "y(x)"] , [style, [lines, 2]] , [gnuplot_preamble, "set grid"] , [nticks, 200])$

(%t4) 

Ahora calcularemos los puntos en los que la derivada es igual a la pendiente de la cuerda:

(%i5) define (Df(x) , diff(f(x),x));

(%o5) $Df(x):=(x-2)(x-1)+(x-3)(x-1)+(x-3)(x-2)$

(%i7) Eq1: Df(x) = 6;   solve([Eq1], [x]), numer;

(%o7) $(x-2)(x-1)+(x-3)(x-1)+(x-3)(x-2)=6$

rat: replaced 9.16515 by 6049/660 = 9.16515

rat: replaced 9.16515 by 6049/660 = 9.16515

(%o8) $[x=0.4725, x=3.52753]$

Finalmente calculamos la ecuación de la recta tangente en estos puntos y hacemos la representación gráfica de estas rectas y la curva anterior:

(%i9) x1 = x1:0.4725;
x2 = x2:3.52753;

(%o9) $x1=0.4725$

(%o10) $x2=3.52753$

(%i11) 'f(x1) = f(x1), numer;
'f(x2) = f(x2), numer;

(%o11) $f(0.4725)=-2.03655$

(%o12) $f(3.52753)=2.03673$

```
(%i13) y2(x):=6*(x - x1) + f(x1);
       y3(x):=6*(x - x2) + f(x2);
(%o13) y2(x):=6 (x-x1)+f(x1)
(%o14) y3(x):=6 (x-x2)+f(x2)
```

```
(%i15) wxplot2d([f(x), y2(x), y3(x)], [x,0,5], [y,-10, 40], [legend, "f(x)" , "y2(x)", "y3(x)"] , [style, [lines, 2]] ,
       [gnuplot_preamble, "set grid"] , [nticks, 200] )$
plot2d: some values were clipped.
```

(%t15)



El teorema del valor medio permite la caracterización de la monotonía de una función derivable en un intervalo. En efecto, si una función $f:[a,b]\to\mathbb{R}$, tal es continua en el intervalo $[a,b]$ y derivable en el intervalo $]a,b[$ entonces, se verifica:

- Si en un intervalo $I=[c,d]\subset[a,b]$ se cumple $Df(x)\geq 0,\ \forall x\in I$, la función es monótona creciente en *I*.
- Si en un intervalo $I=[c,d]\subset[a,b]$ se cumple $Df(x)\leq 0,\ \forall x\in I$, la función es monótona decreciente en *I*.
- Si en un intervalo $I=[c,d]\subset[a,b]$ se cumple $Df(x)=0,\ \forall x\in I$, la función es constante en *I*.

Así mismo, el teorema permite la caracterización de los extremos locales de una función derivable en un intervalo. En efecto, si una función $f:[a,b]\to\mathbb{R}$, tal es continua en el intervalo $[a,b]$ y derivable en el intervalo $]a,b[$ y suponemos que $x_0\in]a,b[$ es un punto crítico de la función, es decir, tal que $Df(x_0)=0$, entonces, se verifica:

- Si existe $r>0$ tal que

$$Df(x)>0,\ \forall x\in]x_0-r,x_0[\quad \text{i} \quad Df(x)<0,\ \forall x\in]x_0,x_0+r[$$

  entonces la función tiene un máximo local en el punto x_0.
- Si existe $r>0$ tal que

$$Df(x) < 0,\ \forall x \in \left]x_0 - r, x_0\right[\quad \text{i} \quad Df(x) > 0,\ \forall x \in \left]x_0, x_0 + r\right[$$

entonces la función tiene un mínimo local en el punto x_0.

Obsérvese que esta ha sido la metodología aplicada para caracterizar los extremos en el apartado anterior.

08-3.- La regla de l'Hôpital

Los contenidos de este apartado se desarrollan en el archivo **Tema_08-3.wxm**.

Si se trata de calcular el límite de un cociente $\lim_{x \to a} \frac{f(x)}{g(x)} = L$ y se cumple que $\lim_{x \to a} f(x) = L_1$ y $\lim_{x \to a} g(x) = L_2$, entonces se verifica:

- Si $L_2 \neq 0$ y $L_2 \neq \infty$ entonces $L = \frac{L_1}{L_2}$
- Si $L_2 = \infty$ y $L_1 \in \mathbb{R}$ entonces $L = 0$
- Si $L_2 = 0$ y $L_1 \neq 0$, entonces $L = \pm\infty$ dependiendo del signo de L_1
- Si los dos límites son cero o infinito (del mismo signo), hay una indeterminación.

Si las funciones son derivables, se puede aplicar el siguiente resultado.

Regla de l'Hôpital (indeterminaciones del tipo 0/0)

Consideremos dos funciones $f, g : [a,b] \subset \mathbb{R} \to \mathbb{R}$. Supongamos que las dos funciones son continuas en $[a,b]$ derivables en $]a,b[$ y que en un punto x_0 de este intervalo se cumple $\lim_{x \to x_0} f(x) = \lim_{x \to x_0} g(x) = 0$ y se quiere estudiar la existencia del límite $\lim_{x \to x_0} \frac{f(x)}{g(x)}$.

Entonces, si la función derivada Dg no se anula en un entorno del punto x_0 y existe $\lim_{x \to x_0} \frac{Df(x)}{Dg(x)} = L$ (que puede ser finito o infinito), entonces se verifica que $\lim_{x \to x_0} \frac{f(x)}{g(x)} = L$.

Ejemplo. Se trata de calcular el límite $\lim_{x \to 0} \frac{\sin(2x)}{3x - x^2}$.

Definimos las funciones, verificamos que se trata de una forma indeterminada y calculamos las funciones derivadas:

```
(%i1) f1(x):=sin(2*x);        f2(x):=3*x-x^2;
(%o1) f1(x):=sin(2 x)
  (%o2) f2(x):=3 x-x^2
```

```
(%i3) 'limit(f1(x), x, 0) = limit(f1(x), x, 0);
      'limit(f2(x), x, 0) = limit(f2(x), x, 0);
```

(%o3) $\lim_{x\to 0} \sin(2x)=0$

(%o4) $\lim_{x\to 0} 3x-x^2=0$

```
(%i5) define ( Df1(x) , diff(f1(x),x) );
      define ( Df2(x) , diff(f2(x),x) );
```

(%o5) $Df1(x):=2\cos(2x)$

(%o6) $Df2(x):=3-2x$

Verificamos que la derivada del denominador no se anula en un entorno del origen y, por lo tanto, se puede aplicar la regla de L'Hôpital, obteniéndose:

```
(%i7) 'limit(Df1(x)/Df2(x), x, 0) = limit(Df1(x)/Df2(x), x, 0);
```

(%o7) $2\left(\lim_{x\to 0} \frac{\cos(2x)}{3-2x}\right)=\frac{2}{3}$

Los algoritmos internos de cálculo de límites de wxMaxima tienen incorporada la metodología adecuada, de modo que se puede calcular el límite directamente:

```
(%i8) 'limit(f1(x)/f2(x), x, 0)=limit(f1(x)/f2(x), x, 0);
```

(%o8) $\lim_{x\to 0} \frac{\sin(2x)}{3x-x^2}=\frac{2}{3}$

Como es sabido, en algunas ocasiones hay que aplicar más de una vez la metodología anterior para llegar a un resultado. Por ejemplo, para calcular el límite $\lim_{x\to 0}\frac{1-\cos(x)}{e^x+e^{-x}-2}$ procedemos a aplicar la metodología:

```
(%i1) f1(x):=1-cos(x);       f2(x):=exp(x)+exp(-x)-2;
```

(%o1) $f1(x):=1-\cos(x)$

(%o2) $f2(x):=\exp(x)+\exp(-x)-2$

```
(%i3) 'limit(f1(x), x, 0) = limit(f1(x), x, 0);
      'limit(f2(x), x, 0) = limit(f2(x), x, 0);
```

(%o3) $\lim_{x\to 0} 1-\cos(x)=0$

(%o4) $\lim_{x\to 0} \%e^{x}+\%e^{-x}-2=0$

```
(%i5) define ( Df1(x) , diff(f1(x),x) );
      define ( Df2(x) , diff(f2(x),x) );
```

(%o5) $Df1(x):=\sin(x)$

(%o6) $Df2(x):=\%e^{x}-\%e^{-x}$

```
(%i7) 'limit(Df1(x), x, 0) = limit(Df1(x), x, 0);
      'limit(Df1(x), x, 0) = limit(Df2(x), x, 0);
```

(%o7) $\lim\limits_{x->0} \sin(x)=0$

(%o8) $\lim\limits_{x->0} \sin(x)=0$

Por lo tanto el límite del cociente de derivadas es también una forma indeterminada. Ahora calculamos las derivadas de segundo orden de las funciones:

```
(%i9) define ( D2f1(x) , diff(f1(x),x,2) );
      define ( D2f2(x) , diff(f2(x),x,2) );
```

(%o9) $D2f1(x):=\cos(x)$

(%o10) $D2f2(x):=\%e^{x}+\%e^{-x}$

Y calculamos los límites de estas funciones en el origen:

```
(%i11) 'limit(D2f1(x), x, 0) = limit(D2f1(x), x, 0);
       'limit(D2f2(x), x, 0) = limit(D2f2(x), x, 0);
```

(%o11) $\lim\limits_{x->0} \cos(x)=1$

(%o12) $\lim\limits_{x->0} \%e^{x}+\%e^{-x}=2$

Por lo tanto, podemos afirmar que se cumple:

$$\lim_{x\to 0}\frac{D^2(1-\cos(x))}{D^2(e^x+e^{-x}-2)}=\frac{1}{2}$$

En efecto:

```
(%i13) 'limit(D2f1(x)/D2f2(x), x, 0) = limit(D2f1(x)/D2f2(x), x, 0);
```

(%o13) $\lim\limits_{x->0} \frac{\cos(x)}{\%e^{x}+\%e^{-x}}=\frac{1}{2}$

Por lo tanto, se puede afirmar que se cumple:

$$\lim_{x\to 0}\frac{D(1-\cos(x))}{D(e^x+e^{-x}-2)}=\frac{1}{2}$$

Y finalmente que:

$$\lim_{x\to 0}\frac{1-\cos(x)}{e^{x}+e^{-x}-2}=\frac{1}{2}$$

En efecto, el resultado es el que da wxMaxima:

```
(%i14) 'limit(f1(x)/f2(x), x, 0) = limit(f1(x)/f2(x), x, 0);
```

$$\text{(\%o14)}\quad \lim_{x\to 0}\frac{1-\cos(x)}{\%e^{x}+\%e^{-x}-2}=\frac{1}{2}$$

No siempre se puede aplicar la regla de L'Hôpital para llegar a un resultado aunque sea con unas cuantas iteraciones, ni wxMaxima da un resultado satisfactorio de forma directa. Hay que recordar que si en una forma indeterminada se aplica la Regla de L'Hôpital con el resultado que el límite del cociente de derivadas no existe, entonces no se debe concluir que el límite inicial no existe. Así, por ejemplo, planteamos el cálculo del límite:

$$\lim_{x\to 0}\frac{x^{2}\sin\left(\frac{1}{x}\right)}{e^{x}-1}$$

Definimos las funciones con wxMaxima y planteamos el cálculo del límite:

```
(%i1) f1(x):=x^2*sin(1/x);  f2(x):=exp(x)-1;
```

$$\text{(\%o1)}\quad f1(x):=x^{2}\sin\left(\frac{1}{x}\right)$$

$$\text{(\%o2)}\quad f2(x):=\exp(x)-1$$

```
(%i3) 'limit(f1(x)/f2(x), x, 0) = limit(f1(x)/f2(x), x, 0);
```

$$\text{(\%o3)}\quad \lim_{x\to 0}\frac{\sin\left(\frac{1}{x}\right)x^{2}}{\%e^{x}-1}=\mathrm{ind}$$

La respuesta de wxMaxima "ind" significa que el programa no puede calcularlo. Si se calculan las derivadas sucesivas y se aplica la regla de L'Hôpital, se va obteniendo una forma indeterminada en la que alguno de los límites no existe. Como sabemos, esto no significa la no existencia del límite. En efecto, en este caso aplicamos la equivalencia de las funciones $f_1(x)=e^{x}-1$ y $f_2(x)=x$ en el origen, con lo que el límite inicial se puede plantear:

$$\lim_{x\to 0}\frac{x^2\sin\left(\frac{1}{x}\right)}{e^x-1}=\lim_{x\to 0}\left(x\sin\left(\frac{1}{x}\right)\right)=0$$

ya que es el producto de un infinitésimo por una función acotada.

La representación gráfica de la función ilustra este hecho.

```
(%i4) wxplot2d([f1(x)/f2(x)], [x,-0.4,0.4], [y, -0.4, 0.4], [ylabel, ""] , [legend, ""] , [style, [lines, 2]] ,
      [gnuplot_preamble, "set grid"], [nticks, 300])$
plot2d: expression evaluates to non-numeric value somewhere in plotting range.
```

(%t4)



Regla de l'Hôpital (indeterminaciones del tipo ∞/∞)

Consideremos dos funciones $f,g:[a,b]\subset\mathbb{R}\to\mathbb{R}$. Supongamos que las dos funciones son continuas en $[a,b]$ derivables en $]a,b[$ y que en un punto x_0 de este intervalo se cumple $\lim_{x\to x_0} f(x)=\lim_{x\to x_0} g(x)=\infty$ y se quiere estudiar la existencia del límite $\lim_{x\to x_0}\frac{f(x)}{g(x)}$.

Entonces, si la función derivada Dg no se anula en un entorno del punto x_0 y existe $\lim_{x\to x_0}\frac{Df(x)}{Dg(x)}=L$ (que puede ser finito o infinito), entonces se verifica que $\lim_{x\to x_0}\frac{f(x)}{g(x)}=L$.

La metodología se aplica de manera idéntica a la indeterminación anterior. Así, por ejemplo, si se trata de calcular el límite

$$\lim_{x\to+\infty} \frac{\log(x^2+1)}{x^3+2x^2+4x+3}$$

observamos en primer lugar que se trata de una indeterminación de este tipo:

```
(%i1) f1(x):=log(x^2 + 1);
      f2(x):=x^3 + 2*x^2 + 4*x + 3;
```

(%o1) $f1(x):=\log(x^2+1)$

(%o2) $f2(x):=x^3+2\,x^2+4\,x+3$

```
(%i3) 'limit(f1(x), x, inf)=limit(f1(x), x, inf);
      'limit(f2(x), x, inf)=limit(f2(x), x, inf);
```

(%o3) $\lim\limits_{x\to\infty} \log(x^2+1)=\infty$

(%o4) $\lim\limits_{x\to\infty} x^3+2\,x^2+4\,x+3=\infty$

Ahora calculamos las funciones derivadas y el límite al infinito de cada una de ellas:

```
(%i5) define ( Df1(x) , diff(f1(x),x) );
      define ( Df2(x) , diff(f2(x),x) );
```

(%o5) $Df1(x):=\dfrac{2\,x}{x^2+1}$

(%o6) $Df2(x):=3\,x^2+4\,x+4$

```
(%i7) 'limit(Df1(x), x, inf) = limit(Df1(x), x, inf);
      'limit(Df2(x), x, inf) = limit(Df2(x), x, inf);
```

(%o7) $2\left(\lim\limits_{x\to\infty}\dfrac{x}{x^2+1}\right)=0$

(%o8) $\lim\limits_{x\to\infty} 3\,x^2+4\,x+4=\infty$

```
(%i9) 'limit(Df1(x)/Df2(x), x, inf) = limit(Df1(x)/Df2(x), x, inf);
```

(%o9) $2\left(\lim\limits_{x\to\infty}\dfrac{x}{(x^2+1)\,(3\,x^2+4\,x+4)}\right)=0$

Este resultado permite llegar a la conclusión final:

$$\lim_{x\to+\infty} \frac{D\log(x^2+1)}{D\left(x^3+2x^2+4x+3\right)}=0 \quad\Rightarrow\quad \lim_{x\to+\infty} \frac{\log(x^2+1)}{\left(x^3+2x^2+4x+3\right)}=0$$

En este caso, el resultado puede ser calculado de forma directa con wxMaxima:

```
(%i10) 'limit(f1(x)/f2(x), x, inf) = limit(f1(x)/f2(x), x, inf);
```

$$(\%o10)\quad \lim_{x\to\infty} \frac{\log(x^2+1)}{x^3+2x^2+4x+3}=0$$

Otras formas indeterminadas. Les indeterminaciones del tipo

$$\infty-\infty,\ 0\cdot\infty,\ 1^{\infty},\ 0^0,\ \infty^0$$

se reducen los casos estudiados mediante transformaciones algebraicas y el uso de las funciones logaritmo neperiano y exponencial.

08-4.- La fórmula de Taylor

Los contenidos de este apartado se desarrollan en el archivo **Tema_08-4.wxm**.

Consideremos una función $f:[a,b]\subset\mathbb{R}\to\mathbb{R}$ y supongamos que es derivable *n* veces en el punto $x_0\in]a,b[$, es decir, que en este punto existen las derivadas de orden superior o sucesivas $Df(x_0),\ D^2f(x_0),\ \ldots,\ D^nf(x_0)$. Se llama polinomio de Taylor de grado *n* de *f* en x_0 al polinomi:

$$T_{n;x_0}(x)=f(x_0)+\frac{Df(x_0)}{1!}(x-x_0)+\frac{D^2f(x_0)}{2!}(x-x_0)^2+\cdots+\frac{D^nf(x_0)}{n!}(x-x_0)^n$$

Es inmediato verificar que este polinomio cumple que su valor y el de sus derivadas de orden superior hasta el orden *n* coinciden con los correspondientes de la función, es decir:

$$T_{n;x_0}(x_0)=f(x_0),\quad D^kT_{n;x_0}(x_0)=D^kf(x_0),\ k=1,2,\ldots,n$$

Además, el polinomio de Taylor $T_n(x)$ de grado *n* de *f* en x_0 cumple la propiedad de ser una aproximación local de grado *n* de la función, es decir:

$$\lim_{x\to x_0}\frac{f(x)-T_{n;x_0}(x)}{\left(x-x_0\right)^n}=0$$

El primer término no nulo del polinomio de Taylor se llama término principal. El polinomio de Taylor aproxima localmente la función en un entorno del punto x_0; la diferencia $R_{n;x_0}(x)=f(x)-T_{n;x_0}(x)$ se denomina *residuo* (o término complementario).

Teorema (de Taylor). Si una función $f:[a,b]\subset\mathbb{R}\to\mathbb{R}$ verifica:

- es continua en el intervalo $[a,b]$
- es derivable sucesivamente en el intervalo $]a,b[$
- las funciones derivadas sucesivas $Df,D^2f,\ldots,D^nf$ son continuas en $]a,b[$
- existe la derivada $D^{n+1}f$ en $]a,b[$

Entonces si $x_0\in[a,b]$, para cada $x\in[a,b]$ existe un punto $t\in]x_0,x[$ tal que:

$$f(x)=T_{n;x_0}(x)+R_{n;x_0}(x)=$$
$$=f(x_0)+\frac{Df(x_0)}{1!}(x-x_0)+\cdots+\frac{D^nf(x_0)}{n!}(x-x_0)^n+\frac{D^{n+1}f(t)}{(n+1)!}(x-x_0)^{n+1}$$

La expresión del residuo que aparece en esta expresión se denomina forma de Lagrange del residuo. La fórmula de Taylor en $x_0 = 0$ se llama fórmula de Maclaurin.

Cabe recordar que el polinomio de Taylor es una aproximación local, válida únicamente en un entorno del punto en el que se calcula el polinomio. La acotación del error cometido la da, como es sabido, el residuo o término complementario.

Observemos en primer lugar que wxMaxima escribe la expresión del polinomio de Taylor de una función de forma general. La sintaxis es muy sencilla y se puede ver a continuación. Así, en un punto x_0:

```
(%i1) taylor(f(x), x, x0, 3);
```

$$(\%o1)/T/\ f(x0)+\left(\left.\frac{d}{d\,x}f(x)\right|_{x=x0}\right)(x-x0)+\frac{\left(\left.\frac{d^2}{d\,x^2}f(x)\right|_{x=x0}\right)(x-x0)^2}{2}+\frac{\left(\left.\frac{d^3}{d\,x^3}f(x)\right|_{x=x0}\right)(x-x0)^3}{6}+\ldots$$

En el origen:

```
(%i2) taylor(f(x), x, 0, 3);
```

$$(\%o2)/T/\ f(0)+\left(\left.\frac{d}{d\,x}f(x)\right|_{x=0}\right)x+\frac{\left(\left.\frac{d^2}{d\,x^2}f(x)\right|_{x=0}\right)x^2}{2}+\frac{\left(\left.\frac{d^3}{d\,x^3}f(x)\right|_{x=0}\right)x^3}{6}+\ldots$$

Veamos ahora unos ejemplos.

Ejemplo 8.4.1.- Se trata de calcular el polinomio de Taylor de grados 1, 2, 3, 4 de la función exponencial en el origen. En primer lugar hay que definir la función y escribir las instrucciones que de wxMaxima que permiten el cálculo de los polinomios de Taylor:

```
(%i1) f1(x):=exp(x);
      taylor(f1(x), x, 0, 1);      taylor(f1(x), x, 0, 2);
      taylor(f1(x), x, 0, 3);      taylor(f1(x), x, 0, 4);
```

$$(\%o1)\ f1(x):=\exp(x)$$

$$(\%o2)/T/\ 1+x+\ldots$$

$$(\%o3)/T/\ 1+x+\frac{x^2}{2}+\ldots$$

$$(\%o4)/T/\ 1+x+\frac{x^2}{2}+\frac{x^3}{6}+\ldots$$

$$(\%o5)/T/\ 1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\ldots$$

Ahora definimos estos polinomios como funciones:

```
(%i6) define ( T1(x) , taylor(f1(x), x, 0, 1) );
      define ( T2(x) , taylor(f1(x), x, 0, 2) );
      define ( T3(x) , taylor(f1(x), x, 0, 3) );
      define ( T4(x) , taylor(f1(x), x, 0, 4) );
```

(%o6)/T/ $T1(x):=1+x+\ldots$

(%o7)/T/ $T2(x):=1+x+\frac{x^2}{2}+\ldots$

(%o8)/T/ $T3(x):=1+x+\frac{x^2}{2}+\frac{x^3}{6}+\ldots$

(%o9)/T/ $T4(x):=1+x+\frac{x^2}{2}+\frac{x^3}{6}+\frac{x^4}{24}+\ldots$

A continuación se trata de observar gráficamente que el polinomio de Taylor es una aproximación local de la función, aproximación que mejora de calidad a medida que se incrementa el grado del polinomio. Para ello haremos unas representaciones gráficas de la función y del polinomio de Taylor de un grado determinado. Así, si dibujamos conjuntamente la función y el polinomio de grado 1 se obtiene:

```
(%i10) wxplot2d([f1(x),T1(x)], [x,-2,2], [ylabel, ""] , [legend, "f(x)" , "T1(x)"] , [style, [lines, 2]] ,
       [gnuplot_preamble, "set grid"], [nticks, 300] )$
```

(%t10)



Ahora hacemos lo mismo con el polinomio de Taylor de grado 2:

```
(%i11) wxplot2d([f1(x),T2(x)], [x,-2,2], [ylabel, ""] , [legend, "f(x)" , "T2(x)"] , [style, [lines, 2]] ,
        [gnuplot_preamble, "set grid"], [nticks, 300] )$
```

(%t11)



Y ahora hacemos lo mismo con el polinomio de Taylor de grado 3:

```
(%i12) wxplot2d([f1(x),T3(x)], [x,-2,2], [ylabel, ""] , [legend, "f(x)" , "T3(x)"] , [style, [lines, 2]] ,
        [gnuplot_preamble, "set grid"], [nticks, 300] )$
```

(%t12)



Como puede observarse, las dos gráficas se van pareciendo cada vez más a medida que se incrementa el grado del polinomio de Taylor. Veamos finalmente la representación gráfica de la función y el polinomio de Taylor de grado 4:

(%i13) wxplot2d([f1(x),T4(x)], [x,-2,2], [ylabel, ""] , [legend, "f(x)" , "T4(x)"] , [style, [lines, 2]] , [gnuplot_preamble, "set grid"], [nticks, 300])$

(%t13) 

En este último caso se observa que dibujo que da el programa casi no permite diferenciar las dos gráficas. Esta dificultad se convierte en casi imposibilidad si el entorno del punto se hace más pequeño:

(%i14) wxplot2d([f1(x),T4(x)], [x,-0.2,0.2], [ylabel, ""] , [legend, "f(x)" , "T4(x)"] , [style, [lines, 2]] , [gnuplot_preamble, "set grid"], [nticks, 300])$

(%t14) 

Ejemplo 8.4.2.- Se trata de hacer lo mismo con la función definida por:

$$f(x) = \cos(x^2),\ x \in \mathbb{R}$$

En primer lugar calculamos el polinomio de Taylor de grado 4:

```
(%i1) f(x):=cos(x^2);
      define (T4(x) , taylor(f(x), x, 0, 4) );
```

(%o1) $f(x):=\cos(x^2)$

(%o2)/T/ $T4(x):=1-\frac{x^4}{2}+\ldots$

Y ahora hacemos la representación gráfica de la función y este polinomio:

```
(%i3) wxplot2d([f(x),T4(x)], [x,-3,3], [y,-2, 2], [ylabel, ""] , [legend, "f(x)" , "T4(x)"] , [style, [lines, 2]] ,
      [gnuplot_preamble, "set grid"], [nticks, 300] )$
plot2d: some values were clipped.
```

(%t3)



Ahora calculamos el polinomio de Taylor de grado 8 y hacemos la representación gráfica conjunta:

```
(%i4) define (T8(x) , taylor(f(x), x, 0, 8) );
```

(%o4)/T/ $T8(x):=1-\frac{x^4}{2}+\frac{x^8}{24}+\ldots$

```
(%i5) wxplot2d([f(x),T8(x)], [x,-3,3], [y,-2,2], [ylabel, ""] , [legend, "f(x)" , "T8(x)"] , [style, [lines, 2]] ,
      [gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300] )$
plot2d: some values were clipped.
```

(%t5)



Finalmente, calculamos el polinomio de Taylor de grado 16 y hacemos la representación gráfica conjunta:

(%i6) define (T16(x) , taylor(f(x), x, 0, 16));

(%o6)/T/ $T16(x) := 1 - \frac{x^4}{2} + \frac{x^8}{24} - \frac{x^{12}}{720} + \frac{x^{16}}{40320} + \ldots$

(%i7) wxplot2d([f(x),T16(x)], [x,-3,3], [y,-2,2], [ylabel, ""] , [legend, "f(x)" , "T16(x)"] , [style, [lines, 2]] , [gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300])$

plot2d: some values were clipped.

(%t7)



Series de potencias.

La propiedad del residuo de la fórmula de Taylor permite escribir:

$$f(x) = \lim_{x \to x_0} T_{n;x_0}(x) = \sum_{k=0}^{+\infty} \frac{D^k f(x_0)}{k!}(x - x_0)^k$$

El último término es una serie de potencias y la igualdad anterior se conoce como desarrollo en serie de potencias de la función. Con wxMaxima se puede obtener la expresión del desarrollo en serie de potencias de una función, tal como se muestra a continuación.

Por ejemplo, para la función exponencial real se tiene:

(%i1) 'exp(x)=powerseries(exp(x), x, 0);

(%o1) $\%e^x = \sum_{i1=0}^{\infty} \frac{x^{i1}}{i1!}$

Una instrucción de wxMaxima permite mejorar la estética de la salida:

```
(%i2) 'exp(x)=niceindices(powerseries(exp(x), x, 0));
```

$$(\%o2)\ \%e^x = \sum_{i=0}^{\infty} \frac{x^i}{i!}$$

Para las funciones seno y coseno:

```
(%i3) 'sin(x)=niceindices(powerseries(sin(x), x, 0));
```

$$(\%o3)\ \sin(x) = \sum_{i=0}^{\infty} \frac{(-1)^i\, x^{2\,i+1}}{(2\,i+1)!}$$

```
(%i4) 'cos(x)=niceindices(powerseries(cos(x), x, 0));
```

$$(\%o4)\ \cos(x) = \sum_{i=0}^{\infty} \frac{(-1)^i\, x^{2\,i}}{(2\,i)!}$$

Para la función arco tangente:

```
(%i5) 'atan(x)=niceindices(powerseries(atan(x), x, 0));
```

$$(\%o5)\ \operatorname{atan}(x) = \sum_{i=0}^{\infty} \frac{(-1)^i\, x^{2\,i+1}}{2\,i+1}$$

También se puede obtener el desarrollo en serie de potencias de una función que no esté predefinida en wxMaxima:

```
(%i6) f4(x):=1/(1+x);     'f4(x)=niceindices(powerseries(f4(x), x, 0));
```

$$(\%o6)\ f4(x) := \frac{1}{1+x}$$

$$(\%o7)\ f4(x) = \sum_{i=0}^{\infty} (-1)^i\, x^i$$

Finalmente dos ejemplos con el logaritmo neperiano, el primero en el origen y el segundo en el punto $x_0 = 1$:

(%i8) 'log(x+1)=niceindices(powerseries(log(x+1), x, 0));

$$(\%o8)\ \log(x+1) = -\sum_{i=1}^{\infty} \frac{(-1)^i x^i}{i}$$

(%i9) 'log(x)=niceindices(powerseries(log(x), x, 1));

$$(\%o9)\ \log(x) = -\sum_{i=1}^{\infty} \frac{(-1)^i (x-1)^i}{i}$$

08-5.- Aplicaciones de la derivada. Optimización

Los contenidos de este apartado se desarrollan en el archivo **Tema_08-5.wxm**.

Caracterización de la convexidad, la concavidad y las inflexiones de unas funciones.

Una función se dice <u>convexa</u> en un intervalo $[x_1, x_2]$ si y sólo si se cumple:

$$f\left(\lambda x_1 + (1-\lambda)x_2\right) \leq \lambda f(x_1) + (1-\lambda) f(x_2), \quad 0 \leq \lambda \leq 1$$

es decir, si el segmento que une los dos extremos $\left(x_1, f(x_1)\right), \left(x_2, f(x_2)\right)$ está por encima de la gráfica de la función. Si la desigualdad es estricta, se acostumbra a decir que la función es estrictamente convexa en el intervalo. Interpretación geométrica de la convexidad:



Figura 8.5.1

Una función se dice <u>cóncava</u> en un intervalo $[x_1, x_2]$ si y sólo si se cumple:

$$f\left(\lambda x_1 + (1-\lambda)x_2\right) \geq \lambda f(x_1) + (1-\lambda) f(x_2), \quad 0 \leq \lambda \leq 1$$

es decir, si el segmento que une los dos extremos $\left(x_1, f(x_1)\right), \left(x_2, f(x_2)\right)$ está por debajo de la gráfica de la función. Si la desigualdad es estricta, se acostumbra a decir que la función es estrictamente cóncava en el intervalo. Interpretación geométrica de la concavidad:



Figura 8.5.2

Un punto en el que la función es cóncava a la izquierda y convexa a la derecha o a la inversa se llama un punto de inflexión. En la Figura 8.5.3 se puede ver un punto de inflexión.



Figura 8.5.3

Una propiedad importante de las funciones derivables es la siguiente: si una función $f:[a,b]\to\mathbb{R}$ es derivable dos veces en el intervalo $]a,b[$ entonces, se verifica:

- si se cumple $Df(x)>0,\ \forall x\in\left]c,d\right[$, la función es convexa en $[c,d]$;
- si se cumple $Df(x)<0,\ \forall x\in\left]c,d\right[$, la función es cóncava en $[c,d]$;
- si la función es derivable tres veces y se cumple $D^2f(x_0)=0,\ D^3f(x_0)\neq 0$, el punto x_0 es un punto de inflexión de f.

Ejemplo 8.5.1.- Se considera la función definida por:

$$f(x) = x^4 - x^3 - 9x^2,\ x \in [-3,3]$$

Veamos su definición con wxMaxima y su representación gráfica:

```
(%i1) f(x):=x^4-x^3-9*x^2;
(%o1) f(x):=x^4-x^3+(-9) x^2
(%i4) wxplot2d([f(x)], [x,-3, 3],  [ylabel, ""] , [legend, "" ] , [style, [lines, 2]] ,
      [gnuplot_preamble, "set grid"], [nticks, 300] )$
```

(%t4)



Calculamos las derivadas de primer y segundo orden de la función:

```
(%i5) define (Df(x) , diff(f(x), x) );
(%o5) Df(x):=4 x^3-3 x^2-18 x
(%i6) define (D2f(x) , diff(f(x), x, 2) );
(%o6) D2f(x):=12 x^2-6 x-18
```

Calculamos ahora las raíces de la derivada de segundo orden:

```
(%i7) Eq1 : D2f(x) = 0 ;
      solve([Eq1], [x]);
(%o7) 12 x^2-6 x-18=0
(%o8) [x=3/2,x=-1]
```

Veamos la representación gráfica de la derivada segunda:

```
(%i9) wxplot2d([D2f(x)], [x,-3, 3],  [ylabel, ""] , [legend, "D2f(x)" ] , [style, [lines, 2]] ,
      [gnuplot_preamble, "set grid"], [nticks, 300] )$
```

(%t9)



Por lo tanto, se puede afirmar que, por ejemplo, la función es convexa en [-3,-2] y la función es cóncava en [0,1]. Estudiamos les raíces de la derivada segunda para ver si son puntos de inflexión, para lo que habrá que calcular la derivada de tercer orden y evaluarla en estos puntos:

```
(%i10) define (D3f(x) , diff(f(x), x, 3) );
```

(%o10) $D3f(x):=24\,x-6$

```
(%i11) x1=x1:-1;       'D3f(x1) = D3f(x1);
```

(%o11) $x1=-1$

(%o12) $D3f(-1)=-30$

```
(%i13) x2=x2:3/2;      'D3f(x2)=D3f(x2);
```

(%o13) $x2=\frac{3}{2}$

(%o14) $D3f\left(\frac{3}{2}\right)=30$

Así pues, en efecto, estos puntos son puntos de inflexión de la función, ya que la derivada de tercer orden de la función en estos puntos no es nula.

Resolución de ecuaciones: el método de Newton – Raphson.

El método de Newton – Raphson, o simplemente de Newton, es un método iterativo para la aproximación de los ceros de una función derivable.

La fórmula de iteración es la siguiente:

$$x_{n+1} = x_n - \frac{f(x_n)}{Df(x_n)},\ n \geq 1$$

La interpretación geométrica es bien conocida: se trata de aproximar el cero mediante las intersecciones de las sucesivas rectas tangentes a la gráfica de la función, como se puede ver en la Figura 8.5.4:



Figura 8.5.4

Ejemplo 8.5.2.- Se trata de resolver la ecuación

$$x^2 - 8\log(x) = 0$$

En primer lugar intentamos resolverla con wxMaxima:

```
(%i1) Eq1:  x^2 - 8*log(x) = 0;
        solve(Eq1, x);
(%o1) x^2-8 log(x)=0
(%o2) [x=-2^(3/2) sqrt(log(x)), x=2^(3/2) sqrt(log(x
```

Vemos que el programa no nos da una respuesta adecuada. Aplicaremos el método de Newton con la función $f(x) = x^2 - 8\log(x)$. En primer lugar debemos definir la función y calcular la función derivada, ya que intervine en la fórmula recurrente:

```
(%i3) f(x) := x^2 - 8*log(x);
```

(%o3) $f(x):=x^2-8\log(x)$

```
(%i4) define (Df(x) , diff(f(x), x) );
```

(%o4) $Df(x):=2\,x-\frac{8}{x}$

Ahora hacemos una representación de la gráfica de la función para saber en qué intervalo se encuentran las soluciones y poder diseñar el procedimiento iterativo correspondiente:

```
(%i5) wxplot2d([f(x)], [x, 0.5, 4],  [y, -2, 6] , [ylabel, ""] , [legend, "f(x)" ] , [style, [lines, 2]] ,
      [gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300] )$
```

(%t5)



Observamos que hay un cero en el intervalo I_1 =]1, 1.5[y otro cero en el intervalo I_2 =]2.5, 3[. A continuación iniciamos el proceso iterativo para calcular el cero en I_2 con $x_1 = 4$:

```
(%i6) x1=x1:4;    x2=x2:x1 - (f(x1))/(Df(x1)), numer;
(%o6) x1 = 4
 (%o7) x2 = 3.181725814826521
(%i8) x3=x3:x2 - (f(x2))/(Df(x2)), numer;
(%o8) x3 = 2.957260178756791
(%i9) x4=x4:x3 - (f(x3))/(Df(x3)), numer;
(%o9) x4 = 2.935049208792494
(%i10) x5=x5:x4 - (f(x4))/(Df(x4)), numer;
(%o10) x5 = 2.934820198893187
(%i11) x6=x6:x5 - (f(x5))/(Df(x5)), numer;
(%o11) x6 = 2.934820174464085
```

Parece que ya se tienen al menos seis decimales exactos y si la precisión se considera suficiente, se puede parar el proceso iterativo. Con wxMaxima hay una instrucción que permite este cálculo de forma directa y con la precisión que se quiera. La sintaxis es la siguiente:

```
(%i12) load(newton1)$
       newton(x^2 - 8*log(x), x, 4, 10^(-5));
(%o13) 2.934820198893187
```

Si se desea una precisión más fina, es decir, una cota menor del error, entonces hay que indicarlo en la instrucción. Así:

```
(%i14) load(newton1)$
       newton(x^2 - 8*log(x), x, 4, 10^(-9));
(%o15) 2.934820174464085
```

Ahora podemos calcular con esta metodología el cero de la función en el intervalo I_1:

```
(%i16) load(newton1)$
       newton(x^2 - 8*log(x), x, 1, 10^(-9));
(%o17) 1.195663759058167
```

Variaciones respecto al tiempo.

En muchos problemas de aplicación de la derivada se plantean situaciones en las que intervienen funciones que dependen del tiempo. A continuación se muestra un ejemplo de la aplicación de la derivada a una de estas situaciones.

Ejemplo 8.5.3.- Se tiene un embudo cónico recto circular que tiene un radio de la base de radio R = 20cm y una altura de H = 40cm. El embudo está situado con el vértice hacia abajo y entra líquido por la parte superior con un caudal de q = 200 cm^3/min. Si inicialmente el embudo está vacío, se trata de calcular la función que expresa el nivel del líquido contenido en el embudo en función del tiempo y razonar que se trata de una función creciente.

Si se designan por r(t), y(t) las funciones que expresan el radio de la base y la altura, respectivamente, del cono que forma el líquido dentro del embudo en el instante t, entonces el volumen de líquido contenido en el embudo es

```
(%i1) V(t):=(%pi/3)*r(t)^2*y(t);
```

(%o1) $V(t):=\frac{\pi}{3}r(t)^2\,y(t)$

De la geometría del problema deducimos que:

(%i2) r(t):=(R/H)*y(t);   'V(t) = V(t);

(%o2) $r(t):=\frac{R}{H}\,y(t)$

(%o3) $V(t)=\frac{\pi\, y(t)^3\, R^2}{3\, H^2}$

Por lo tanto, igualando este volumen al que se sabe que hay dentro del embudo como producto del caudal de entrada por el tiempo, resulta:

(%i4) Eq22: q*t=V(t);

(%o4) $q\,t=\frac{\pi\, y(t)^3\, R^2}{3\, H^2}$

De esta ecuación podemos deducir la expresión de la función altura de líquido:

(%i5) (y(t))^3 = (3*q*H^2) / (%pi*R^2) * t;

(%o5) $y(t)^3=\frac{3\,q\,t\,H^2}{\pi\, R^2}$

(%i6) define (y(t) , ((3*q*H^2)/(%pi*R^2))^(1/3) * t^(1/3));

(%o6) $y(t):=\frac{3^{1/3}\,q^{1/3}\,t^{1/3}\,H^{2/3}}{\pi^{1/3}\,R^{2/3}}$

Si designamos por *K* el coeficiente que multiplica a la raíz cúbica del tiempo, entonces podemos expresar la función altura de líquido en el embudo:

(%i7) q=q:200$ R=R:20$ H=H:40$

(%i10) 'K=K:((3*q*H^2)/(%pi*R^2))^(1/3), numer;

(%o10) $K=9.141562994681664$

(%i11) define (y(t) , K*t^(1/3));

(%o11) $y(t):=9.141562994681664\,t^{1/3}$

La representación de la gráfica de esta función es:

```
(%i12) wxplot2d([y(t)], [t, 0,100], [ylabel, ""] , [legend, "y(t)" ] , [style, [lines, 2]] ,
       [gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300] )$
```

(%t12)



El crecimiento se puede observar en la gráfica, sin embargo se puede analizar esta característica de la función viendo el signo de la función derivada:

```
(%i13) define (Dy(t) , diff(y(t), t) );
```

(%o13) $Dy(t):=\frac{3.047187664893888}{t^{2/3}}$

Dado que esta derivada es positiva en todo punto, se puede concluir el crecimiento de la función. Ahora se puede razonar que la velocidad a la que va aumentando el nivel del líquido en el interior del embudo es decreciente con el tiempo. En primer lugar haremos la representación gráfica de esta derivada:

```
(%i14) wxplot2d([Dy(t)], [t, 0.1,90] , [y, 0, 8] , [ylabel, ""] , [legend, "y(t)" ] , [style, [lines, 2, 2]] ,
       [gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300] )$
plot2d: some values were clipped.
```

(%t14)



Ahora verificaremos que la derivada de la altura es decreciente calculando la derivada de segundo orden y viendo que es negativa:

(%i15) diff(Dy(t), t);

(%o15) $-\frac{2.031458443262592}{t^{5/3}}$

Finalmente, se puede hacer algún cálculo como por ejemplo calcular la velocidad a la que aumenta el nivel de líquido en el embudo cuando la altura de líquido es de 25cm, para lo cual determinaremos en qué instante se alcanza esta altura y después calcularemos la derivada de la altura en este instante:

(%i17) Eq24: y(t)=25;   solve([Eq24], t);

(%o17) $9.1416\, t^{1/3}=25$

rat: replaced 9.1416 by 8072/883 = 9.1416

(%o18) $[t=\frac{10757271671875}{525948789248}]$

(%i19) t0 = t0 : 10757271671875/525948789248, numer;

(%o19) $t0=20.453$

(%i20) 'Dy(t0)=Dy(t0);

(%o20) $Dy(20.453)=0.407$

Optimización.

Una de las aplicaciones más significativas de la derivada es la determinación de extremos de una función, problemas que se conocen como optimización. Veamos un ejemplo.

Ejemplo 8.5.4.- Se considera un círculo de radio R; se trata de determinar el sector circular que hay que extraer para que con lo que queda de círculo se forme un cono de volumen máximo.

Designamos por *x* el ángulo (en radianes) del sector que se extrae. Calculamos la longitud del arco de este sector y de la parte restante:

(%i1) L1(x):=R*x;   L2(x):=2*%pi*R - R*x;

(%o1) $L1(x):=R\, x$

(%o2) $L2(x):=2\,\pi\, R-R\, x$

El radio de la base del cono que se forma con lo que queda de círculo es:

(%i3) r(x):=(2*%pi*R - R*x)/(2*%pi);

(%o3) $r(x):=\frac{2\,\pi\, R-R\, x}{2\,\pi}$

Por el teorema de Pitágoras calculamos la altura de este cono:

(%i4) h(x):=sqrt(R^2 - ((2*%pi*R - R*x)/(2*%pi))^2);

$$\text{(\%o4) } h(x):=\sqrt{R^2-\left(\frac{2\,\pi\,R-R\,x}{2\,\pi}\right)^2}$$

Y ahora podemos calcular su volumen:

(%i5) V(x):=(%pi/3)*(r(x)^2)*h(x);

$$\text{(\%o5) } V(x):=\frac{\pi}{3}\,r(x)^2\,h(x)$$

Está claro que esta función está definida en el intervalo [0, 2π]. Para determinar el máximo de esta función calculamos la función derivada:

(%i6) diff(V(x), x);

$$\text{(\%o6) } \frac{R\,(2\,\pi\,R-x\,R)^3}{48\,\pi^3\,\sqrt{R^2-\frac{(2\,\pi\,R-x\,R)^2}{4\,\pi^2}}}-\frac{R\,(2\,\pi\,R-x\,R)\,\sqrt{R^2-\frac{(2\,\pi\,R-x\,R)^2}{4\,\pi^2}}}{6\,\pi}$$

Parece razonable intentar simplificar esta expresión:

(%i7) ratsimp(%);

$$\text{(\%o7) } -\frac{(2\,x^3-12\,\pi\,x^2+16\,\pi^2\,x)\,R^4+(x^3-6\,\pi\,x^2+12\,\pi^2\,x-8\,\pi^3)\,R^4}{24\,\pi^2\,\sqrt{4\,\pi\,x-x^2}\,|R|}$$

Ahora podemos definirla como función:

(%i8) DV(x):=-((2*x^3-12*%pi*x^2+16*%pi^2*x)*R^4+
(x^3-6*%pi*x^2+12*%pi^2*x-8*%pi^3)*R^4)/(24*%pi^2*sqrt(4*%pi*x-x^2)*R);

$$\text{(\%o8) } DV(x):=\frac{-((2\,x^3-12\,\pi\,x^2+16\,\pi^2\,x)\,R^4+(x^3-6\,\pi\,x^2+12\,\pi^2\,x+(-8)\,\pi^3)\,R^4)}{24\,\pi^2\,\sqrt{4\,\pi\,x-x^2}\,R}$$

Para calcular los puntos críticos igualamos a cero y resolvemos:

(%i9) Eq1:DV(x)=0;

$$\text{(\%o9) } \frac{-(2\,x^3-12\,\pi\,x^2+16\,\pi^2\,x)\,R^4-(x^3-6\,\pi\,x^2+12\,\pi^2\,x-8\,\pi^3)\,R^4}{24\,\pi^2\,\sqrt{4\,\pi\,x-x^2}\,R}=0$$

(%i10) solve([Eq1], [x]);

(%o10) $[x=-\frac{(2\sqrt{6}-6)\pi}{3}, x=\frac{(2\sqrt{6}+6)\pi}{3}, x=2\pi]$

Ahora calculamos numéricamente estas soluciones:

```
(%i11) x1=x1:-((2*sqrt(6)-6)*%pi)/3, numer;
       x2=x2:((2*sqrt(6)+6)*%pi)/3, numer;
       x3=x3:2*%pi, numer;
```

(%o11) $x1=1.152985986532131$

(%o12) $x2=11.41338462782704$

(%o13) $x3=6.283185307179586$

La segunda no pertenece al intervalo y, por lo tanto, se puede descartar. La tercera es precisamente el extremo superior del intervalo, en el que la función volumen es nula. Por lo tanto hay que considerar sólo la primera y para determinar la tipología del extremo hay que calcular la derivada segunda en este punto:

(%i14) diff(DV(x), x);

(%o14) $$\frac{-(6x^2-24\pi x+16\pi^2)R^4-(3x^2-12\pi x+12\pi^2)R^4}{24\pi^2\sqrt{4\pi x-x^2}\,R}-\frac{(4\pi-2x)(-(2x^3-12\pi x^2+16\pi^2x)R^4-(x^3-6\pi x^2+12\pi^2x-8\pi^3)R}{48\pi^2(4\pi x-x^2)^{3/2}R}$$

```
(%i15) D2V(x):=(-9*x^2+36*%pi*x-28*%pi^2)/(24*%pi^2*sqrt(4*%pi*x-x^2)) -
       ((4*%pi-2*x)*(-3*x^3+18*%pi*x^2-28*%pi^2*x+8*%pi^3))/(48*%pi^2*(4*%pi*x-x^2)^(3/2));
```

(%o15) $$D2V(x):=\frac{(-9)x^2+36\pi x+(-28)\pi^2}{24\pi^2\sqrt{4\pi x-x^2}}-\frac{(4\pi-2x)((-3)x^3+18\pi x^2+(-28)\pi^2x+8\pi^3)}{48\pi^2(4\pi x-x^2)^{3/2}}$$

(%i16) 'D2V(x1)=D2V(x1), numer;

(%o16) $D2V(1.152985986532131)=-0.18377629847393$

Per lo tanto la función es cóncava y se trata de un máximo local. Este extremo es absoluto ya que el volumen es nulo en los extremos del intervalo.

Para ilustrar gráficamente el problema se hace una representación gráfica de la función volumen con un radio unidad:

(%i22) wxplot2d([V(x)], [x, 0, 2*%pi] , [ylabel, ""] , [legend, "V(x)"] , [style, [lines, 2, 3]] , [gnuplot_preamble, "set size ratio 1; set grid"], [nticks, 300])$

(%t22)